// Evol functions.
// Authors:
//    Qwerty Dragon
//    Reid
// Description:
//    Random sailor dialogues between two categories of NPCs.

function	script	sailorfood	{
    mesn;

    .@q = rand(0, 400) / 100;
    if (.@q == 0) goto L_RandomA;
    if (.@q == 1) goto L_RandomB;
    if (.@q > 1) goto L_RandomC;

L_RandomA:
    mesq l("Hey.");
    next;
    mesq l("What did Gugli say about the box? Was it ok?");
    next;

    menu
        l("It's ok."), L_Fine,
        l("He needs more food."), -;

    mes "";
    mesn;
    mesq l("Oh really? I'll put more food in the next box then.");

    close;

L_Fine:
    mes "";
    mesn;
    mesq l("It's alright! Just one more box and it'll be ok.");

    close;

L_RandomB:
    mesq l("Thanks for the help!");
    next;
    mesq l("These boxes are way too heavy to be lifted by only one person, all the way onto the ship.");

    close;

L_RandomC:
    mesq l("I think I'll be done soon, since I almost have a box full of @@s!", getitemlink(CrocClaw));
    next;
    mesq l("And you? How's it going on your side?");
    next;

    return;
}
